Important Loop, Tuner, & Phaser Design Considerations (Q Demystified) - a DX Engineering discussion & program

Mark Connelly — WA1ION DX Labs — 29 JAN 1985

Often technical articles in the DX press mention a parameter known as "Q"; seldom is there any mathematical description of what this quantity is - the DXer reading the article usually does get the idea that Q has something to do with selectivity, be it of a receiver, a loop antenna, or of a preselector / tuner used on random length wires. Q is also mentioned with regards to coils: a coil with higher Q allows greater selectivity and gain when used in a tuned circuit than does a coil with low Q. "Q spoiling" is also mentioned, chiefly with regards to tuned circuits employed in phasing systems. This may confuse some - (why, after all, should we want to spoil Q? .. doesn't that reduce gain & selectivity?) Hopefully, when you finish this article and get a chance to run the BASIC program contained therein, some of these questions will be answered.

What is Q, anyway?

The letter Q, derived from Quality Factor, is applied to a parameter which describes the efficiency of a tuned LC (inductor / capacitor) tank circuit, an assemblage of several such circuits (as in a filter), or a single inductive element to be used in a tuned circuit. Capacitors use a similar loss-describing parameter (D, Dissipation Factor) but, as capacitors are generally several orders of magnitude less lossy than inductors in tank circuits used at frequencies below 30 MHz, their loss contributions won't be dealt with in this article.

Q of a single inductor

Q of a single inductor is calculated by Q = (F * L) / (R * 159.15495), where F is in kHz, L is in microhenries (uH), and R is in ohms. Conversely, effective resistance of an inductor may be calculated by R = (F * L) / (Q * 159.15495). A J. W. Miller # 9250-224 (220 uH molded inductor), with a listed Q of 55 at 790 kHz would therefore have an effective series resistance of about 20 ohms. Coils used for loop antennae generally have a considerably higher Q than the small molded inductors used in tuner bandswitches.

Q of a parallel-tuned LC circuit

The need for the tank Q parameter arises from the fact that components used in tuned circuits are not purely reactive (inductive or capacitive) - resistance (both series & parallel) may be present. Furthermore, loads imposed upon the tuned circuit may also possess resistive impedance characteristics. The greater the amount of loss caused by resistance, the lower the Q of the tank in question will be; tuning sharpness will decrease.

Resonance in an LC tuned circuit occurs at $F = (159154.95) / \sqrt{(L*C)}$ where F is in kHz, L is in microhenries (uH), and C is in picofarads (pF). A parallel-tuned LC circuit, such as a loop antenna tuned with a variable capacitor, if comprised of perfect components, should have infinite gain at the one frequency to which it is tuned and zero signal present at all other frequencies.

Of course, no such loop exists. Gain which is finite and bandwidth which is not infinitesimally small describe real-life conditions. Many factors contribute to resistive losses, but the two major contributors (at frequencies below 30 MHz) are effective series resistance of the inductive element and effective parallel resistance of the load. Note the use of the word "effective".

T664-2

Effective series resistance of the coil in question is derived from actual DC resistance of the winding and from core losses - the latter part of this effective resistance may be somewhat frequency-dependent, but, for the purposes of this introductory article, we'll consider it to be constant over the frequency range of interest. Effective series resistance of a coil can be calculated if the coil Q at a frequency near that of operation is specified (as in the case of the previously-mentioned 220 uH Miller miniature molded inductor).

Effective parallel load resistance is the actual load resistance in the case of a high-impedance load such as a FET amp. input directly fed from the "tank high" and "tank low"(e. g. ground) points or a multiplied-up value of the load's resistance if the load is loosely-coupled to the tank by means of a tap on the tank coil, a link coil near the tank coil, a small-value coupling capacitor, or a large-value coupling inductor.

Q is a selectivity measurement which is defined as the ratio of the resonant frequency of a tuned circuit to the 3 dB bandwidth of that circuit. The RLCQ (BASIC) program takes as operator input the effective shunt (parallel) resistance contributed by the load, the effective series resistance of the coil, the value of the inductor (in uH), and either the frequency or the value of the capacitor. Output generated includes a swept-frequency table giving the magnitude of impedance in ohms (to which signal voltage is proportional), ratio of complex to resistive impedance, required tuning capacitance at each frequency given, and phase shift from the zero-degree reference phase at resonance. Other output information includes 3 dB bandwidth, phase shift caused by a 5 pF shift in tuning capacitance from resonance (an important parameter for phasing unit design), resonant frequency shift caused by a 5 pF shift from resonance, and, of course, the Q for the circuit whose characteristics were inputted.

Output is in a write-to-diskfile format; the program may be modified for direct transferral to paper. Three sample output files are reproduced after the program listing. In each case, a very-high-Q inductance (such as would be used on a loop antenna) of 220 uH is used. The first file uses a 330K load impedance, such as one would have at the input of a FET "front end card": the tuning is very sharp and phase angle changes very rapidly with any shift in the tuning capacitance. The first file demonstrates a desirable situation for a loop antenna or wire tuner whose output is connected directly to a receiver or to a broadband amplifier ahead of a receiver; conversely, this case is undesirable for a tuner used in a phasing system because the change in phase angle per unit change in capacitance is so great that adjustment required to produce a null is exceedingly touchy and narrow-banded - nulling attempts are virtually useless. On the other hand, the only detriment to such high Q in a standalone active-loop is the possibility of the FET amp. going into oscillation. Reducing the gate-to-ground resistor to about 100K usually provides just enough Q-spoiling to stop undesired regeneration without wreaking havoc with gain & selectivity. The second and third files demonstrate (at 600 kHz and 1200 kHz respectively) the characteristics of the tank circuit of the first output file when loaded by a 15K Q-spoiling resistor. 15K has been established as a value which works well in phasing applications: it allows enough Q-spoiling to make null achievement manageable and null bandwidth sufficient without reducing signal level excessively and without reducing phase-shift-per-capacitance-shift to a point that not enough shift for nulls can be produced with modest variations in tuning capacitance.

The program and sample outputs which follow should help the loop, tuner, and phaser designer to simulate many tuned circuits without the need to build them first and without the need to make hours of exhaustive measurements.

(program listing) (DEC PDP-11 BASIC-PLUS-2 V2.2 - 00)

```
1 DIM G(10),FX(10)
2 STAR$=" ****************************************************************"
5 P=3.1415927\K=180/P
6 E=(2^.5)/2\EL=E-.0001\EH=E+.0001
7 PRINT"PROGRAM TITLE = RLCQ.B2S"\PRINT"Parallel R-L-C Tank Analyser"\PRINT
8 FL=0\FH=0
9 INPUT "DATA OUTPUT FILE TITLE";S$\OPEN S$ FOR OUTPUT AS FILE #1%
10 PRINT #1,STAR$\PRINT #1\PRINT #1,"  This is file: ";S$\PRINT #1
14 INPUT "(min. = 25) shunt R, ohms";R1
15 IF R1<25 THEN PRINT"Illegal value"\GO TO 14
18 INPUT "(max. = 500) series R of inductor, ohms";R2
19 IF R2>500 THEN PRINT"Illegal value"\GO TO 18
20 INPUT "(min. = .1 / max. = 50000) L, uH";L
21 IF L>50000 OR L<.1 THEN PRINT"Illegal value"\GO TO 20
22 ANSWER=0\INPUT "Resonant freq. in kHz if known (else enter 0)";ANSWER
23 IF ANSWER<>0 THEN C=25330/(((ANSWER/1000)^2)*L)\GO TO 25
24 INPUT "(min. = 10 / max. = 1000)  C, pF";C
25 IF C<10 OR C>1000 THEN PRINT"Illegal value"\GO TO 22
26 FOR I=1 TO 3
27 IF I=1 THEN C=C-5
28 IF I=2 THEN C=C+10\REM ORIGINAL C + 5
29 IF I=3 THEN C=C-5 \REM ORIGINAL C
32 LA=L*1E-6\ CA= C*1E-12
33 FR=1/((2000*P)*((LA*CA)^.5))
35 IF I=1 THEN XFH=FR
36 IF I=2 THEN XFL=FR
37 NEXT I
38 PRINT #1,"  shunt R = ";R1;" ohms / coil R = ";R2;" ohms"
39 PRINT #1,"  L = ";L;" uH / C = ";C;" pF"
40 PRINT #1,"  Resonant Frequency = ";FR;" kHz"
41 F=FR\GOSUB 900\R=A1
42 F=XFL\GOSUB 600\XBL=B
44 F=XFH\GOSUB 600\XBH=B
50 SB=0\S=0
100 FOR N=1 TO 10
105 IF SB=0 THEN F=FR*(1-(((10-N)^2)/101.25))
110 IF SB=2 THEN F=FR*((((N-1)^2)/20.25)+1)
120 GOSUB 600\ REM MATH
160 G(N)=A/R\FX(N)=F
170 IF N=1 AND SB=2 THEN 300
180 IF S<>0 THEN 210
185 FO$=        " #######.##    #######.##    #.###   #####.#     ####.###"
190 PRINT #1
192 PRINT #1,"    F, kHz    MAG.Z, ohms   MAG Z/R    CE,pF    PHASE, DEG"
200 PRINT #1," ____________________________________________________________"
210 S=1
250 PRINT #1% USING FO$,F,A,A/R,CE,B
300 NEXT N
301 SB=SB+1
302 IF SB=1 THEN GOSUB 700\SB=SB+1
303 IF SB=2 THEN 100
304 IF SB=3 THEN GOSUB 800\GO TO 1000
500 REM  >>>>>          Parallel Impedance SUBROUTINE            <<<<<
510 US=(U1*U2)-(V1*V2)
520 VS=(V1*U2)+(V2*U1)
530 UU=U1+U2\VV=V1+V2
540 DE=(UU^2)+(VV^2)
550 IF DE=0 THEN DE=1E-9
560 U3=((US*UU)+(VS*VV))/DE
570 V3=((VS*UU)-(US*VV))/DE
580 RETURN
600 REM   >>>>>                 Math SUBROUTINE                   <<<<<
610 W=F*2000*P
615 U1=R1\V1=0
620 U2=0.0001\V2=-1/(W*CA)
625 GOSUB 500
```

T66-4-3

```
630 U1=U3\V1=V3
635 U2=R2\V2=W*LA
638 GOSUB 500
640 IF U3=0 THEN U3=.0001\ REM PREVENT DIVIDE-BY-ZERO ERROR
650 A=((U3^2)+(V3^2))^.5\ REM MAGNITUDE OF IMPEDANCE
670 B=(ATN(V3/U3))*K\ REM PHASE ANGLE IN DEGREES
675 CE=25330/(((F/1000)^2)*L)\REM EQUIV. RESONATING C IN PF
690 RETURN
700 REM    >>>>>        -3 dB Low Side Point SUBROUTINE           <<<<<
710 FOR M=1 TO 9
715 IF G(1)>EH THEN GL=1\GO TO 797 ELSE GL=0
717 IF G(M)>EL AND G(M)<EH THEN FL=FX(M)\GO TO 795
720 IF G(M)<E AND G(M+1)>E THEN F1=FX(M)\F2=FX(M+1)\GO TO 730
725 NEXT M
730 F=(F1+F2)/2
740 GOSUB 600
750 IF A/R>EH THEN F2=F\GO TO 730
760 IF A/R<EL THEN F1=F\GO TO 730
770 FL=F
795 BL=B
797 RETURN
800 REM    >>>>>        -3 dB High Side Point SUBROUTINE          <<<<<
810 FOR M=1 TO 9
815 IF G(10)>EH THEN GH=1\GO TO 897 ELSE GH=0
817 IF G(M)>EL AND G(M)<EH THEN FH=FX(M)\GO TO 895
820 IF G(M)>E AND G(M+1)<E THEN F1=FX(M)\F2=FX(M+1)\GO TO 830
825 NEXT M
830 F=(F1+F2)/2
840 GOSUB 600
850 IF A/R>EH THEN F1=F\GO TO 830
860 IF A/R<EL THEN F2=F\GO TO 830
870 FH=F
895 BH=B
897 RETURN
900 REM    >>>>>             Peakfind SUBROUTINE                  <<<<<
910 GOSUB 600
920 A1=A
930 F=F-.01\GOSUB 600\REM TEST FOR INCREASING MAG. Z @ DECREASING F
940 IF A<A1 THEN 990\REM NO SHIFT IN PEAK
945 A1=A
950 F=F-10\GOSUB 600\REM 10 kHz DOWNWARD INCREMENT
955 IF A>A1 THEN A1=A\GO TO 950
960 F=F+.001\GOSUB 600\REM 'HOME IN' ON SOLUTION
965 IF A>A1 THEN A1=A\GO TO 960
990 RETURN
999 REM   *********     PROGRAM OUTPUT CONCLUSION     *********
1000 BW=FH-FL\PRINT #1
1002 TXT$="  Average Phase Shift (+/- 5 pF from resonance) =  "
1003 PRINT #1,TXT$;(XBL-XBH)/2;" deg"
1004 TXT$="  Avg.Res.Freq. Shift (+/- 5 pF from resonance) =  "
1005 PRINT #1,TXT$;(XFH-XFL)/2;" kHz"
1006 PRINT #1
1007 IF GL=1 OR GH=1 THEN 1009
1008 PRINT #1,"            3 dB Bandwidth  =  ";BW;" kHz"
1009 IF GL=1 THEN 1011
1010 PRINT #1,"    Low Side -3 dB point  =  ";FL;" kHz";"  Phase = ";BL;" deg"
1011 IF GH=1 THEN 1013
1012 PRINT #1,"   High Side -3 dB point  =  ";FH;" kHz";"  Phase = ";BH;" deg"
1013 IF GL=1 OR GH=1 THEN 1020
1014 PRINT #1,"            Tank Circuit Q  =  ";FR/BW
1020 S=0\PRINT #1\PRINT #1,STAR$\CLOSE #1%
1025 PRINT\INPUT"(1=YES) MORE CALCULATIONS";S
1030 IF S=1 THEN PRINT\PRINT\PRINT\GO TO 7
1050 END
```

(end of program listing; output files follow)

****

This is file: 0622033OK.TXT

shunt R = 330000 ohms / coil R = 1 ohms
L = 220 uH / C = 319.823 pF
Resonant Frequency = 600.003 kHz

F, kHz	MAG.Z, ohms	MAG Z/R	CE,pF	PHASE, DEG
120.00	172.79	0.001	7995.5	89.610
220.74	352.99	0.002	2362.9	89.722
309.63	583.35	0.003	1200.9	89.716
386.67	914.13	0.004	770.1	89.658
451.85	1442.92	0.006	563.9	89.538
505.19	2398.93	0.011	451.1	89.302
546.67	4447.34	0.020	385.3	88.781
576.30	10274.10	0.046	346.7	87.287
594.08	41058.80	0.184	326.2	79.321
600.00	223006.00	1.000	319.8	-0.025
629.63	8593.74	0.039	290.4	-87.857
718.52	2287.99	0.010	223.0	-89.470
866.67	1102.70	0.005	153.3	-89.764
1074.08	673.47	0.003	99.8	-89.866
1340.75	464.11	0.002	64.0	-89.912
1666.68	343.04	0.002	41.4	-89.937
2051.86	265.21	0.001	27.3	-89.952
2496.31	211.57	0.001	18.5	-89.962
3000.02	172.79	0.001	12.8	-89.969

Average Phase Shift (+/- 5 pF from resonance) = 76.6183 deg
Avg.Res.Freq. Shift (+/- 5 pF from resonance) = 4.6908 kHz

3 dB Bandwidth = 2.23169 kHz
Low Side -3 dB point = 598.889 kHz Phase = 44.9352 deg
High Side -3 dB point = 601.121 kHz Phase = -45.0722 deg
Tank Circuit Q = 268.856

****

****

This is file: 0622015K.TXT

shunt R = 15000 ohms / coil R = 1 ohms
L = 220 uH / C = 319.823 pF
Resonant Frequency = 600.003 kHz

F, kHz	MAG.Z, ohms	MAG Z/R	CE,pF	PHASE, DEG
120.00	172.77	0.012	7995.5	88.980
220.74	352.77	0.024	2362.9	88.435
309.63	582.84	0.040	1200.9	87.591
386.67	912.28	0.062	770.1	86.330
451.85	1435.82	0.098	563.9	84.295
505.19	2367.16	0.161	451.1	80.638
546.67	4255.63	0.290	385.3	73.073
576.30	8418.71	0.573	346.7	54.934
594.08	13844.60	0.943	326.2	19.351
600.00	14679.80	1.000	319.8	-0.002
629.63	7423.97	0.506	290.4	-59.686
718.52	2261.13	0.154	223.0	-81.197
866.67	1099.67	0.075	153.3	-85.752
1074.08	672.79	0.046	99.8	-87.412
1340.75	463.88	0.032	64.0	-88.220
1666.68	342.95	0.023	41.4	-88.686
2051.86	265.16	0.018	27.3	-88.985
2496.31	211.55	0.014	18.5	-89.191
3000.02	172.78	0.012	12.8	-89.339

T66-4-4

Average Phase Shift (+/- 5 pF from resonance) = 15.4684 deg
Avg.Res.Freq. Shift (+/- 5 pF from resonance) = 4.6908 kHz

3 dB Bandwidth = 33.8992 kHz
Low Side -3 dB point = 583.314 kHz Phase = 44.9282 deg
High Side -3 dB point = 617.213 kHz Phase = -45.068 deg
Tank Circuit Q = 17.6996

****

****

This is file: 1222015K.TXT

shunt R = 15000 ohms / coil R = 1 ohms
L = 220 uH / C = 79.9558 pF
Resonant Frequency = 1200.01 kHz

F, kHz	MAG.Z, ohms	MAG Z/R	CE,pF	PHASE, DEG
240.00	345.46	0.023	1998.9	88.500
441.48	704.95	0.047	590.7	87.198
619.26	1163.06	0.078	300.2	85.462
773.34	1814.51	0.122	192.5	82.960
903.71	2832.97	0.190	141.0	79.008
1010.38	4566.82	0.306	112.8	72.134
1093.34	7638.92	0.512	96.3	59.162
1152.60	12073.40	0.809	86.7	35.938
1188.16	14684.90	0.984	81.6	10.121
1200.01	14918.70	1.000	80.0	-0.001
1259.27	11273.00	0.756	72.6	-40.952
1437.05	4375.59	0.293	55.8	-72.973
1733.34	2181.83	0.146	38.3	-81.614
2148.16	1341.53	0.090	25.0	-84.860
2681.50	926.44	0.062	16.0	-86.455
3333.35	685.35	0.046	10.4	-87.379
4103.73	530.08	0.036	6.8	-87.974
4992.62	422.97	0.028	4.6	-88.384
6000.04	345.48	0.023	3.2	-88.680

Average Phase Shift (+/- 5 pF from resonance) = 29.3848 deg
Avg.Res.Freq. Shift (+/- 5 pF from resonance) = 37.613 kHz

3 dB Bandwidth = 133.418 kHz
Low Side -3 dB point = 1135.18 kHz Phase = 44.9662 deg
High Side -3 dB point = 1268.6 kHz Phase = -45.0271 deg
Tank Circuit Q = 8.99434

****

CIRCUIT MODEL



This is file AMIDON.DAT

M CONNELLY
30 WILLIAM RD
BILLERICA MA 01866

Amidon Toroidal Core Data
========================

FT-114-72 u=2000 AL = 1270 mH / 1000T $1.70
(use for 100 - 500 kHz inductors)

L (uH)	Turns
10000	89
4700	60
2200	41
1000	28
470	19
220	13

****

T-106-1 u=20 AL = 325 uH / 100T $1.65
(use for 475 - 1100 kHz inductors)

L (uH)	Turns
470	120
220	82
100	55
47	38

****

T-106-2 u=10 AL = 135 uH / 100T $1.65
(use for 1100 - 3500 kHz inductors)

L (uH)	Turns
100	86
47	59
22	40
10	27

****

T-50-6 u=8 AL = 40 uH / 100T $0.60
(use for 3000 - 7500 kHz inductors)

L (uH)	Turns
22	74
10	50
4.7	34
2.2	23

****

FT-82-77 u=2000 AL = 1060 mH / 1000T $0.90
(use for 100 - 2500 kHz transformers)

****

FT-82-43 u=850 AL = 557 mH / 1000T $0.90
(use for 2000 - 7500 kHz transformers)

****

Inductances were calculated by the following program:

```
  5 PRINT"COIL.B2S"
 10 PRINT\INPUT"(1=IRON, 2=FERRITE)  CORE MATERIAL";CORE
 20 IF CORE<>1 AND CORE<>2 THEN 10
 30 PRINT\INPUT"AL";AL
 40 PRINT\INPUT"INDUCTANCE, UH";L\PRINT
 50 IF CORE=1 THEN PRINT"IRON CORE"\T=100*((L/AL)^.5)\GO TO 100
 60 IF CORE=2 THEN PRINT "FERRITE CORE"\T=1000*((L/(1000*AL))^.5)
100 PRINT "AL = ";AL;"  * L = ";L;" UH  *  # TURNS = ";T
110 Q=0\PRINT\INPUT"(SAME CORE, NEW L) (1=YES) RUN AGAIN";Q
120 IF Q=1 THEN 40
200 END
```